Average impact of multiple conversion errors Imagine four people stand in a line. The first looks at a thermometer, and tells the temperature to the person on their right. This continues down the line, until the last person writes the temperature on a sheet of paper. Unfortunately, due to a miscommunication, every person in the line converts the temperature to their preferred unit, without knowing which it already was.
Let's say the people's preferred units are celsius, fahrenheit, fahrenheit, celsius. The first reads the temperature as 20°C. The second converts to fahrenheit, and determines that it is 68°F. The third, not knowing they have received the temperature in °F already, determines that it is 154.4°F. Finally, it is converted back to °C by the fourth person, who determines that it is 68°C.
Task:
You will be given two inputs: a temperature (t), and a number of people (n). Your program (or function) should return the average difference between the original and resulting temperatures, for every possible assortment of n people.
There are two temperature units: °F and °C. To convert between them, you can use \$F=\frac95C+32\$ and \$C=\frac59(F-32)\$.
You can choose whether your program receives the inputted temperature in °F or °C. The first person should always receive an accurate temperature, converted to their preferred unit only if necessary.
As an example, we'll use an inputted temperature of 41°F, and just 3 people. There are 8 temperature preferences possible:
FFF   41°F  -&gt;  41.0  -&gt;  105.8  -&gt;  222.44  -&gt;  222.44°F
FFC   41°F  -&gt;  41.0  -&gt;  105.8  -&gt;   41.00  -&gt;  105.80°F
FCF   41°F  -&gt;  41.0  -&gt;    5.0  -&gt;   41.00  -&gt;   41.00°F
FCC   41°F  -&gt;  41.0  -&gt;    5.0  -&gt;  -15.00  -&gt;    5.00°F
CFF   41°F  -&gt;   5.0  -&gt;   41.0  -&gt;  105.80  -&gt;  105.80°F
CFC   41°F  -&gt;   5.0  -&gt;   41.0  -&gt;    5.0  -&gt;    41.00°F
CCF   41°F  -&gt;   5.0  -&gt;  -15.0  -&gt;    5.00  -&gt;    5.00°F
CCC   41°F  -&gt;   5.0  -&gt;  -15.0  -&gt;  -26.11  -&gt;  -15.00°F

The average distance from 41°F is 54.88°F, which is the program's output.
I/O:
Temperature should be represented as a float, decimal, or fraction of some sort. The accuracy should be within reasonable bounds; for inputs less than ten people at low temperatures there shouldn't be floating point errors or imprecision noticeable in the first four decimal digits.
You may represent the temperatures inputted and outputted as either °F or °C, but must be consistent (although the input can output can be different units as long as they don't change). Temperatures can be negative. The number of people will never be less than two.
The output is referred to as a temperature above, although it's technically the arithmetic mean of multiple temperatures.
Test Cases:
Input unit is the same as output unit for all test cases.
41°F   2   -&gt;  25.2
41°F   3   -&gt;  54.88
41°F   4   -&gt;  77.236444...
41°F   10  -&gt;  295.4268...
20°C   4   -&gt;  57.21218...
-1°C   2   -&gt;  12.133...
-20°C  2   -&gt;  6.2222...
-40°F  4   -&gt;  0

Other:
This is code-golf, shortest answer in bytes per language wins!
APL (Dyalog Unicode), 41 bytes

(|40+⎕)+.×(|÷≢)1-1.8*1(-,+)⍨⍣(⎕-2)⊢1-3|⍳4

Try it online!
A full program. Takes the number of people, then the initial temperature. TIO version is wrapped in a function for easier testing. As I pointed out in a comment, the initial and final units do not matter as long as they are equal.
The math
The code exploits the fact that Fahrenheit-Celsius conversion has a fixpoint at -40 (-40°F = -40°C). Then we can add 40 to all temperature readings, and then the whole series of conversion reduces into a (positive or negative) integer power of \$\frac95\$.
$$
F+40=\frac95(C+40), \quad C+40 = \frac59(F+40)
$$
Since every temperature value is offset by the constant value of 40, the difference between two offset temperature values is the same as that of the original. Therefore, it suffices to apply the offset to the initial temperature and ignore it to the end.
Let's call \$r = \frac95\$.
Now to the part of counting all conversion paths. Let's say the initial reading is in Fahrenheit. Then:

Initial conversion: F→F (\$r^0\$) or F→C (\$r^{-1}\$).
Final conversion: F→F (\$r^0\$) or C→F (\$r^1\$).
All inner conversions: C→F (\$r^1\$) or F→C (\$r^{-1}\$).

So the task reduces to finding all the individual terms in the expansion of \$(r^0 + r^{-1})(r^1+r^{-1})^{n-2}(r^1+r^0)\$. Then plug in the value of \$r\$, evaluate the average of differences to 1, and then multiply with the initial temperature (with offset).
The code
(|40+⎕)+.×(|÷≢)1-1.8*1(-,+)⍨⍣(⎕-2)⊢1-3|⍳4

⊢1-3|⍳4  ⍝ A fancy way to create an array of 0 ¯1 1 0
         ⍝ which represents the powers in (1+r^-1)(r+1)
⍣(⎕-2)   ⍝ Repeat n-2 times...
1(-,+)⍨  ⍝ multiply the given polynomial with (r+r^-1)
         ⍝ and list all powers
1.8*     ⍝ Plug in r = 1.8 and evaluate the powers
1-       ⍝ Difference of each value from 1
(|÷≢)    ⍝ Average of absolute differences from 1
+.×      ⍝ Product and sum with...
(|40+⎕)  ⍝ Absolute difference between -40 and t